

# **PB-700 ENCORE DES COURBES**

■ L'étude d'une fonction mathématique est tout de même plus aisée en présence de sa courbe représentative. Avec le PB-700 et un programme adapté, on pourra la voir à l'écran ou sur le papier de l'imprimante. Auparavant, le programme aura étudié la fonction entre deux limites fixées, calculé l'échelle et placé les axes.

Les commentaires du programme n'apparaissent pas dans la liste sous forme de REM, mais ils peuvent être détaillés. A la ligne 15, se situe le calcul du champ d'étude de la fonction et celui du pas. Si les bornes d'étude sont respectivement négative et positive, cette ligne calcule aussi la position de l'axe des ordonnées. Si les deux bornes sont négatives, la longueur d'étude est la valeur absolue de la borne inférieure de l'axe des X (ligne 20), auquel cas on ne voit que la partie négative de l'axe des abscisses, tracée à droite de l'écran ( $V = 159$ ). Si les deux bornes sont positives, l'axe est à gauche de l'écran ( $V = 0$ , ligne 25), la longueur d'étude correspondant à la plus grande valeur des X, S.

La boucle FOR...NEXT (lignes 35 à 60) étudie la fonction point par point, recherchant son maximum (ligne 50) et son minimum (ligne 55). Le calcul de l'échelle s'effectue à la ligne 70, 32 étant le nombre de points de la verticale de l'écran du PB-700.

S'il n'y a pas de valeur négative de X, l'axe se trouve en bas de l'écran ( $W = 31$ , ligne 75) ; s'il n'y a pas de valeur positive, l'axe des X se trouve en haut de l'écran ( $W = 0$ , ligne 80).

Le repère est affiché (ligne 85), puis la fonction est tracée point par point (lignes 100 à 150). La ligne 160 attend qu'une touche soit frappée, elle empêche un terrible READY P0 de venir tout effacer.

Avant d'utiliser ce programme, il faut introduire, à la ligne 200, l'équation de la fonction dont on souhaite la courbe représentative. Après avoir entré les bornes d'étude de cette fonction, le repère s'affiche puis la courbe apparaît. L'appui sur RETURN donne alors l'échelle.

Enfin, il faudra faire attention à l'ensemble de définition de la fonction dont on souhaite la courbe : si l'intervalle d'étude introduit n'est pas inclus dans cet ensemble, un message d'erreur (erreur mathématique) est inévitable. C'est notamment le cas si  $X = 0$ , avec la fonction  $Y = 1/X$ .

*Olivier DUFALLY*

**Courbes**

Programme pour PB-700

Auteur Olivier Dufailly

Copyright LIST et l'auteur

```
1  REM COURBE UNIVERSELLE
5  CLEAR :CLS :W=16
10 INPUT "Petit X";R,"Grand X";S
15 D=ABSR+ABSS:U=D/160:IF R<0 THEN IF S>0
   THEN U=R-U:GOTO 35
20 IF ABSR>ABSS THEN D=-R:U=159:GOTO 30
25 D=S:U=0
30 U=D/160
35 FOR X=R TO S STEP U
40 GOSUB 200
50 IF Y>A THEN A=Y
55 IF B<Y THEN B=Y
60 NEXT X
70 E=(ABSA+ABSB)/32
75 IF B=0 THEN W=31
80 IF A=0 THEN W=0
85 CLS :DRAW(0,W)-(159,W):DRAW(U,0)-(U,31)
100 FOR X=R TO S STEP U
110 GOSUB 200
130 Y=W-INT(Y/E)
140 IF Y>=0 THEN IF Y<32 THEN DRAW(INT(X/U+U),Y)
150 NEXT X
160 IF INKEY$="" THEN 160
170 PRINT "Ech 1/";E;:END
200 Y=SIN(2*X)
210 RETURN
```